

## The physics angles $\theta^*$ and $\phi^*$ for SANE with Maple V

The physics system for asymmetry calculation is defined as  
 $z$  = along incident electron momentum  $\vec{k}$ , i.e. along beam  
 $y$  = along cross product  $\vec{k} \times \vec{k}'$  (points UP in lab, for  $\vec{k}'$  horizontal and pointing towards BETA, to the left of the beam)  
 $x$  = normal to  $y, z$  (points towards BETA for SANE)

We want to find the angles  $\theta^*$  and  $\phi^*$  of the target spins relative to the momentum transfer vector  $\vec{q} = \vec{k} - \vec{k}'$ . For this purpose, we apply Euler angle rotations to the lab coordinate system into the  $\vec{q}$  system.

First, rotate counterclockwise about  $z$  by the azimuthal scattering angle  $\phi = \phi_e$  (There are no subscripts in Maple V, so we use plain theta and phi for  $\theta \equiv \theta_q$  and  $\phi \equiv \phi_e$ ):

```
Rzphi:=linalg[matrix] ([[cos(phi),sin(phi),0], [-sin(phi),cos(phi),0], [0,0,1]])
```

$$Rzphi := \begin{bmatrix} \cos(\phi) & \sin(\phi) & 0 \\ -\sin(\phi) & \cos(\phi) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

For  $\phi = \pi/2$  his rotation takes the  $x$  axis into the initial  $y$  axis.

Then, rotate about the physics  $y$  by the angle  $\theta_q$  between  $\vec{k}$  and  $\vec{q}$  from  $\vec{k}$  to  $\vec{q}$ .

This is a CLOCKWISE rotation, since  $\vec{q}$  is to the right of  $\vec{k}$  and  $y$  is UP, so sines are  $*(-1)$ . This rotation is identical to RSS, because the SANE and RSS systems are just related by a  $\pi$  rotation about  $\vec{k}$ , so the relative directions of  $\vec{k}$  and  $\vec{q}$  stay the same.

```
> Ryq:=linalg[matrix] ([[cos(theta),0,sin(theta) ], [0,1,0], [-sin(theta),0,cos(theta)]]);
```

$$Ryq := \begin{bmatrix} \cos(\theta) & 0 & \sin(\theta) \\ 0 & 1 & 0 \\ -\sin(\theta) & 0 & \cos(\theta) \end{bmatrix}$$

Final rotation, about the rotated  $z^*$  axis

```
Rzg:=linalg[matrix] ([[cos(gamma),sin(gamma),0 ], [-sin(gamma),cos(gamma),0], [0,0,1]])
```

$$Rzg := \begin{bmatrix} \cos(\gamma) & \sin(\gamma) & 0 \\ -\sin(\gamma) & \cos(\gamma) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> with(linalg):
> Rstar:=multiply(Rzg,Ryq,Rzphi);
```

*Rstar* :=

$$\begin{bmatrix} \cos(\gamma) \cos(\theta) \cos(\phi) - \sin(\gamma) \sin(\phi), & \cos(\gamma) \cos(\theta) \sin(\phi) + \sin(\gamma) \cos(\phi), & \cos(\gamma) \sin(\theta) \\ -\sin(\gamma) \cos(\theta) \cos(\phi) - \cos(\gamma) \sin(\phi), & -\sin(\gamma) \cos(\theta) \sin(\phi) + \cos(\gamma) \cos(\phi), & -\sin(\gamma) \sin(\theta) \\ -\sin(\theta) \cos(\phi), & -\sin(\theta) \sin(\phi), & \cos(\theta) \end{bmatrix}$$

Since there is no third rotation, i.e.  $\gamma=0$ ,

```
Rzg1:=linalg[matrix](3,3,[1,0,0,0,1,0,0,0,1])
```

$$Rzg1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> Rstar1:=evalm(Rzg1&*Ryq&*Rzphi);
```

$$Rstar1 := \begin{bmatrix} \cos(\theta) \cos(\phi) & \cos(\theta) \sin(\phi) & \sin(\theta) \\ -\sin(\phi) & \cos(\phi) & 0 \\ -\sin(\theta) \cos(\phi) & -\sin(\theta) \sin(\phi) & \cos(\theta) \end{bmatrix}$$

PARA:

B points along  $-z$  in all systems

```
> Bpara:=linalg[matrix](3,1,[0,0,-Bz]);
```

$$Bpara := \begin{bmatrix} 0 \\ 0 \\ -Bz \end{bmatrix}$$

```
> Bstarpara:=evalm(Rstar1 &* Bpara);
```

$$Bstarpara := \begin{bmatrix} -\sin(\theta) Bz \\ 0 \\ -\cos(\theta) Bz \end{bmatrix}$$

```
> cosstar:=Bstarpara[3,1]/Bz;
```

$$cosstar := -\cos(\theta)$$

```
> phistar:=arctan(Bstarpara[2,1]/Bstarpara[1,1] );
```

$$phistar := 0$$

$\phi^* = 0$  or  $180$ , depending on the sign of the denominator. Here denominator is negative, so  $\phi^* = 180$

$80^\circ$ : B points almost along  $x$  in physics system.

```
> B80:=linalg[matrix] (3,1,[Bf*sin(beta),0,Bf*cos(beta)]);
```

$$B80 := \begin{bmatrix} Bf \sin(\beta) \\ 0 \\ Bf \cos(\beta) \end{bmatrix}$$

```
> Bstar80:=evalm(Rstar1 &* B80);
```

$$Bstar80 := \begin{bmatrix} \cos(\theta) \cos(\phi) Bf \sin(\beta) + \sin(\theta) Bf \cos(\beta) \\ -\sin(\phi) Bf \sin(\beta) \\ -\sin(\theta) \cos(\phi) Bf \sin(\beta) + \cos(\theta) Bf \cos(\beta) \end{bmatrix}$$

```
> cosstar:= Bstar80[3,1]/Bf:cosStar:=simplify(cosstar);
```

$$\cosStar := -\sin(\theta) \cos(\phi) \sin(\beta) + \cos(\theta) \cos(\beta)$$

```
> phistar:=arctan(Bstar80[2,1]/Bstar80[1,1]): phiStar:=simplify(phistar);
```

$$\phiStar := -\arctan\left(\frac{\sin(\phi) \sin(\beta)}{\cos(\theta) \cos(\phi) \sin(\beta) + \sin(\theta) \cos(\beta)}\right)$$

Sample result

Recall  $\theta = \theta_q, \phi =$  out-of-plane  $\phi_e$  for substitutions.  $\beta$  is field direction in the lab's (not always same as physics)  $x$ - $z$  plane.

```
> thstar:=evalf(arccos( subs(theta=22.29*Pi/180,phi=(360-0)*Pi/180,
beta=80*Pi/180,cosstar))/Pi*180);
```

$$thstar := 102.290000$$

```
> phis:= subs(theta=22.29*Pi/180,phi=(360-0)*Pi/180,beta=90*Pi/180,phiStar):
```

```
> phis1:=evalf(phis*180./Pi);
```

$$phis1 := 0$$

So  $\cos \theta^* = -\sin \theta_q \cos \phi_e \sin \beta + \cos \theta_q \cos \beta$ ,  
and  $\tan \phi^*$  is in fourth quadrant ( $-$  sine,  $+$  cosine).